
	Other comments:
	


New functions and definitions
	  const integer tsc_TimeUnitsPerFrame_Ts := 307200;                         /* Ts: Basic time unit for LTE; Ts = 1/(15000 * 2048); 307200 * Ts = 10ms (frame duration; Tf) */

  const integer tsc_TimeUnitsPerFrame_Tc := tsc_TimeUnitsPerFrame_Ts * 64;  /* Tc: Basic time unit for NR;  Tc = 1/(480000 * 4096); see TS 38.211 clause 4.1 and 3.2 */

  const integer tsc_TimeUnitsPerSubframe_Tc := tsc_TimeUnitsPerFrame_Tc / 10;  

	  function f_SystemTimeUnits_ConvertFromCellTimingInfo(CellTimingInfo_Type p_CellTimingInfo) return integer

  {

    var integer v_SystemTimeUnits;

    v_SystemTimeUnits := (p_CellTimingInfo.SfnOffset + (1024 * p_CellTimingInfo.HsfnOffset)) * tsc_TimeUnitsPerFrame_Tc;

    v_SystemTimeUnits := v_SystemTimeUnits + (p_CellTimingInfo.Tcell * 64);

    if (ispresent(p_CellTimingInfo.TcOffset)) {    // NR only

      v_SystemTimeUnits := v_SystemTimeUnits + p_CellTimingInfo.TcOffset;

    }

    return v_SystemTimeUnits;

  }

	  function f_SystemTimeUnits_ConvertFromSubFrameTiming(SubFrameTiming_Type p_SubFrameTiming) return integer

  {

    var integer v_SystemTimeUnits;

    var SlotOffset_Type v_SlotOffset;

    if (ischosen(p_SubFrameTiming.HSFN.Any) or ischosen(p_SubFrameTiming.SFN.Any) or ischosen(p_SubFrameTiming.Subframe.Any) or ischosen(p_SubFrameTiming.Slot.Any)) {

      FatalError(__FILE__, __LINE__, "cannot convert SubFrameTiming into SystemTimeUnits");

    }

    v_SystemTimeUnits := (p_SubFrameTiming.Subframe.Number + (10 * (p_SubFrameTiming.SFN.Number + (1024 * p_SubFrameTiming.HSFN.Number)))) * tsc_TimeUnitsPerSubframe_Tc;

    if (ischosen(p_SubFrameTiming.Slot.SlotOffset)) {

      // NOTE: if p_SubFrameTiming.Slot.FirstSlot or p_SubFrameTiming.Slot.SlotOffset.Numerology0 then no slot needs to be counted

      v_SlotOffset := p_SubFrameTiming.Slot.SlotOffset;

      select union (v_SlotOffset) {

        case (Numerology1) { v_SystemTimeUnits := v_SystemTimeUnits + (v_SlotOffset.Numerology1 * (tsc_TimeUnitsPerSubframe_Tc / 2)); }

        case (Numerology2) { v_SystemTimeUnits := v_SystemTimeUnits + (v_SlotOffset.Numerology2 * (tsc_TimeUnitsPerSubframe_Tc / 4)); }

        case (Numerology3) { v_SystemTimeUnits := v_SystemTimeUnits + (v_SlotOffset.Numerology3 * (tsc_TimeUnitsPerSubframe_Tc / 8)); }

        case (Numerology4) { v_SystemTimeUnits := v_SystemTimeUnits + (v_SlotOffset.Numerology4 * (tsc_TimeUnitsPerSubframe_Tc / 16)); }

      }

    }

    return v_SystemTimeUnits;

  }

	  function f_SystemTimeUnits_ConvertToSubFrameTiming(integer p_SystemTimeUnits,

                                                     SlotTimingInfo_Type p_SlotNumerology) return SubFrameTiming_Type

  { 
    var SubFrameTiming_Type v_SubFrameTiming;

    var SlotOffset_Type v_SlotOffset;

    var integer v_SystemTimeUnits := p_SystemTimeUnits;

    var integer v_HyperSFNs;

    var integer v_SFNs;

    var integer v_Subframes;

    var integer v_Remainder;

    if (v_SystemTimeUnits < 0) {

      v_SystemTimeUnits := v_SystemTimeUnits + (1024 * 1024 * tsc_TimeUnitsPerFrame_Tc);

    }

    v_Remainder := v_SystemTimeUnits mod tsc_TimeUnitsPerSubframe_Tc;

    v_Subframes := v_SystemTimeUnits / tsc_TimeUnitsPerSubframe_Tc; 

    if (not ischosen(p_SlotNumerology.SlotOffset)) {

      v_SubFrameTiming.Slot.FirstSlot := true;

      v_Subframes := v_Subframes + ((v_Remainder + (tsc_TimeUnitsPerSubframe_Tc / 2)) / tsc_TimeUnitsPerSubframe_Tc);     /* v_Subframes rounded up or down; may not be needed but kept for backward compatibility */

    } else {

      select union (p_SlotNumerology.SlotOffset) {

        case (Numerology0) { v_SubFrameTiming.Slot.SlotOffset.Numerology0 := true; }

        case (Numerology1) { v_SubFrameTiming.Slot.SlotOffset.Numerology1 := v_Remainder mod (tsc_TimeUnitsPerSubframe_Tc / 2); }

        case (Numerology2) { v_SubFrameTiming.Slot.SlotOffset.Numerology2 := v_Remainder mod (tsc_TimeUnitsPerSubframe_Tc / 4); }

        case (Numerology3) { v_SubFrameTiming.Slot.SlotOffset.Numerology3 := v_Remainder mod (tsc_TimeUnitsPerSubframe_Tc / 8); }

        case (Numerology4) { v_SubFrameTiming.Slot.SlotOffset.Numerology4 := v_Remainder mod (tsc_TimeUnitsPerSubframe_Tc / 16); }

      }

    } 

    v_SFNs := v_Subframes / 10;

    v_HyperSFNs := v_SFNs / 1024;

    v_SubFrameTiming.Subframe.Number := v_Subframes mod 10;

    v_SubFrameTiming.SFN.Number := v_SFNs mod 1024;

    v_SubFrameTiming.HSFN.Number := v_HyperSFNs mod 1024;

    return v_SubFrameTiming;
  }


f_CellTimingInfo_TimingOtherCell
	  function f_CellTimingInfo_TimingOtherCell(CellTimingInfo_Type p_CellTimingInfo1,

                                            template (value) TimingInfo_Type p_TimingInfo1,

                                            CellTimingInfo_Type p_CellTimingInfo2) return template (value) TimingInfo_Type

  {































    var integer v_SystemTimeUnits;

    var SubFrameTiming_Type v_SubFrameTiming;

    if (not f_TimingInfo_IsNow(p_TimingInfo1)) {

      return cs_TimingInfo_Now;

    }

    v_SubFrameTiming := valueof(p_TimingInfo1.SubFrame);

    v_SystemTimeUnits := f_SystemTimeUnits_ConvertFromSubFrameTiming(v_SubFrameTiming)     // SystemTimeUnits for cell 1 

      + f_SystemTimeUnits_ConvertFromCellTimingInfo(p_CellTimingInfo1)                     // add SFN offset and TCell of cell 1 (=> normalised time) @sic R5s120289: offset needs to be added but not substracted sic@

      - f_SystemTimeUnits_ConvertFromCellTimingInfo(p_CellTimingInfo2);                    // subtract SFN offset and TCell of cell 2                 @sic R5s120289: offset needs to be substracted but not added sic@

    v_SubFrameTiming := f_SystemTimeUnits_ConvertToSubFrameTiming(v_SystemTimeUnits, v_SubFrameTiming.Slot);

    return cs_TimingInfo_NR(v_SubFrameTiming);

  }

